Maple-assisted proof of formula for A295201

Robert Israel

18 November 2017

There are $2^8 = 256$ possible configurations for a 2×4 sub-array. Consider the 256×256 transition matrix T such that $T_{ij} = 1$ if the bottom two rows of a 3×4 sub-array could be in configuration i while the top two rows are in configuration j (i.e. the middle row is compatible with both i and j, and each 1 in that row is horizontally or vertically adjacent to 2 or 3 1's), and 0 otherwise. The following Maple code computes it. I'm encoding a configuration

$$\begin{bmatrix} b_1 & b_2 & b_3 & b_4 \\ b_5 & b_6 & b_7 & b_8 \end{bmatrix}$$

as $b+1$ where $b_1 b_2 b_3 b_4 b_5 b_6 b_7 b_8$ is the binary representation of b. The $+1$ is needed because matrix indices start at 1 rather than 0.

```
> q:= proc(a,b) local r,s,t,M,i;
    s:= floor((a-1)/16);
    if s <> (b-1) mod 16 then return 0 fi;
    s:= convert(s+16,base,2);
    r:= convert(16+floor((b-1)/16),base,2);
    t:= convert(16+ ((a-1) mod 16),base,2);
    M:= Vector(4);
    for i from 1 to 3 do if s[i] = 1 and s[i+1] = 1 then M[i]:= M
 [i]+1; M[i+1]:= 1 fi od;
    for i from 1 to 4 do if s[i]=1 then
      M[i]:= M[i]+r[i]+t[i];
      if M[i] <= 1 or M[i]=3 then return 0 fi;
    fi od;
    1
  end proc:
  T:= Matrix(256,256, q);
```

$$T := \begin{bmatrix} \textit{256 x 256 Matrix} \\ \textit{Data Type: anything} \\ \textit{Storage: rectangular} \\ \textit{Order: Fortran\_order} \end{bmatrix} \quad \textbf{(1)}$$

Thus $a(n) = u\, T^n\, v$ where u and v are row and column vectors respectively with $u_i = 1$ for i corresponding to configurations with bottom row $(0, 0, 0, 0)$, 0 otherwise, and $v_i = 1$ for i corresponding to configurations with top row $(0, 0, 0, 0)$, 0 otherwise. The following Maple code produces these vectors.

```
> u:= Vector[row](256):
  v:= Vector(256):
  for i from 0 to 15 do u[16*i+1]:= 1; v[i+1]:= 1;
  od:
```

To check, here are the first few entries of our sequence.

```
> seq(u . T^n . v, n = 1 .. 10);
```

$$1, 4, 14, 42, 108, 284, 777, 2146, 5887, 16061 \quad (2)$$

Now here is the minimal polynomial P of T, as computed by Maple.

```
> P:= unapply(LinearAlgebra:-MinimalPolynomial(T, t), t);
```

$$P := t \to t^{43} - 8\,t^{42} + 28\,t^{41} - 63\,t^{40} + 101\,t^{39} - 118\,t^{38} + 118\,t^{37} - 93\,t^{36} + 85\,t^{35} + 7\,t^{34} - 111\,t^{33} + 310\,t^{32} - 361\,t^{31} + 333\,t^{30} - 231\,t^{29} + 240\,t^{28} - 236\,t^{27} - 196\,t^{26} + 550\,t^{25} - 1106\,t^{24} - 81\,t^{23} + 358\,t^{22} - 666\,t^{21} - 513\,t^{20} + 667\,t^{19} - 504\,t^{18} - 945\,t^{17} + 741\,t^{16} + 261\,t^{15} + 618\,t^{14} + 1136\,t^{13} + 214\,t^{12} - 99\,t^{11} - 140\,t^{10} - 224\,t^{9} - 97\,t^{8} - 9\,t^{7} + 10\,t^{6} + 15\,t^{5} + 7\,t^{4} + t^{3} \quad (3)$$

This turns out to have degree 43. Thus we will have $0 = u\,P(T)\,T^n\,v = \sum_{i=0}^{43} p_i\,a(i+n)$ where p_i is the coefficient of t^i in $P(t)$. That corresponds to a homogeneous linear recurrence of order 43, which would hold true for any u and v. It seems that with our particular u and v we have a recurrence of order only 13, corresponding to a factor of P.

```
> factor(P(t));
```

$$t^3\,(t-1)\,(t+1)\,(t^9 - t^8 - 2\,t^7 - 4\,t^6 - 7\,t^5 - 13\,t^4 - 11\,t^3 - 5\,t^2 - 4\,t - 1)\,(t^{10} - 2\,t^9 + 3\,t^8 - 2\,t^7 + 5\,t^6 - 4\,t^5 + 2\,t^4 + 2\,t^3 + t^2 + 3\,t + 1)\,(t^6 - 2\,t^5 + 3\,t^4 - 3\,t^3 + 2\,t^2 + t - 1)\,(t^{13} - 3\,t^{12} + 2\,t^{11} - 2\,t^{10} - 5\,t^9 + t^8 + 3\,t^7 + t^6 + 6\,t^5 + 10\,t^4 + t^3 - t^2 - t - 1) \quad (4)$$

```
> Q:= unapply(t^13-3*t^12+2*t^11-2*t^10-5*t^9+t^8+3*t^7+t^6+6*
  t^5+10*t^4+t^3-t^2-t-1, t);
```

$$Q := t \to t^{13} - 3\,t^{12} + 2\,t^{11} - 2\,t^{10} - 5\,t^9 + t^8 + 3\,t^7 + t^6 + 6\,t^5 + 10\,t^4 + t^3 - t^2 - t - 1 \quad (5)$$

The complementary factor $R(t) = \frac{P(t)}{Q(t)}$ has degree 30.

```
> R:= unapply(normal(P(t)/Q(t)), t);
```

$$R := t \to (-1 - 6\,t + t^{27} - 5\,t^{26} + 11\,t^{25} - 18\,t^{24} + 20\,t^{23} - 26\,t^{22} + 21\,t^{21} - 25\,t^{20} + 12\,t^{18} - 53\,t^{17} + 43\,t^{16} - 69\,t^{15} + 29\,t^{14} - 70\,t^{13} + 45\,t^{12} + 61\,t^{11} - 137\,t^{10} + 60\,t^9 + 3\,t^8 - t^7 + 90\,t^6 + 22\,t^5 - 2\,t^4 + 3\,t^3 - 8\,t^2)\,t^3 \quad (6)$$

Now we want to show that $b(n) = u\,Q(T)\,T^n\,v = 0$ for all n. This will certainly satisfy the order-30 recurrence

$$\sum_{i=0}^{30} r_i\,b(i+n) = \sum_{i=0}^{30} r_i u\,Q(T)\,T^{n+i} v = u\,Q(T)\,R(T)\,T^n\,v = u\,P(T)\,T^n\,v = 0$$

where r_i are the coefficients of $R(t)$. To show all $b(n) = 0$ it suffices to show $b(0) = ... = b(29) = 0$.

This would take some time to do naively, so it's worthwhile to do some pre-calculation.

```
> w:= u . Q(T):
  V[0]:= v:
  for i from 1 to 29 do V[i]:= T . V[i-1] od:
  seq(w . V[i],i=0..29);
```

$$0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0 \quad (7)$$